

```

        .text
        .global push
        .type push, %function
        @ void push(int elemento, ELEM **coda)
        @ r0 elemento r1 coda
push:    push {r4-r6,lr} @ salvataggio registri e parametri
        mov r4, r0      @ elemento da inserire
        mov r5, r1      @ puntatore a coda
        @ allocazione area di memoria tramite malloc
        mov r0, #8      @ sizeof int+punt = 8 byte
        bl malloc
        mov r6, r0      @ indirizzo nuovo elemento
        @ riempiamo i campi del nuovo elemento
        str r4, [r6]    @ info = elemento
        ldr r3, [r5]    @ puntatore a coda
        str r3, [r6,#4] @ next = coda
        @ quindi aggiustiamo il nuovo valore del puntatore alla coda
        str r6, [r5]    @ coda = nuovo elemento
        pop {r4-r6, pc} @ quindi restituiamo il controllo al chiamante


        .global isEmpty
        .type isEmpty, %function
        @ int isEmpty(ELEM **coda)
        @ r0 coda
isEmpty: ldr r0, [r0]    @ carichiamo la cella puntata da coda
        cmp r0, #0      @ se NULL vale 0
        moveq r0, #1    @ e quindi isEmpty = true (cioè 1)
        movne r0, #0    @ altrimenti sEmpty = false (cioè 0)
        mov pc, lr      @ restituisci il controllo al chiamante


        .global pop
        .type pop, %function
        @ int pop(ELEM **coda)
        @ r0 coda
pop:     push {r4,r5,lr} @ salvataggio registri
        ldr r3, [r0]    @ puntatore al primo elemento
        cmp r3, #0      @ confronto con NULL
        beq nonfine    @ coda è vuota
        ldr r4, [r3]    @ altrimenti recupera coda→info
        ldr r5, [r3,#4] @ e coda→next
        str r5, [r0]    @ aggiusta il puntatore a coda = vecchio coda→next
        mov r0, r3      @ puntatore all'elem da eliminare
        bl free         @ rilascia l'area di memoria
        mov r0, r4      @ restituisci il valore info salvato
        pop {r4,r5,pc} @ restituisci il controllo


nonfine: @ se la lista era vuota (punt = NULL)
        sub r3, #1      @ il valore da restituire è -1
        pop {r4,r5,pc} @ restituisci il controllo

```